

# De Mozart

*De Mozart à Amstrad : en fait, ce choix n'est pas tout-à-fait un hasard, tout du moins en ce qui me concerne. Lorsque l'on connaît un peu la vie de ce cher Wolfgang Amadeus, on en arrive très vite à supposer que si Mozart vivait actuellement, il serait*

## DUO (MOZART)

*Menuetto*

(Köchel 487)

244

Entrons maintenant plus sérieusement dans le vif du sujet et voyons tout ce qu'il va falloir faire depuis le moment où vous allez choisir une partition jusqu'au moment où vous écouterez votre œuvre assis au fond de votre canapé.

Tout d'abord le choix de la partition. Un premier critère, le plus important, va conditionner votre choix : le nombre de voix. Attention, l'Amstrad ne peut jouer que sur trois voix simultanément alors, les fugues de Bach, c'est raté. Un deuxième critère dépend de votre niveau musical. Certaines fois, on trouve des morceaux où les voix ont tendance à se croiser et il n'est pas toujours facile de les différencier. Alors, si vous êtes

débutant, prenez plutôt des partitions où tout vous semble clair. L'expérience (malheureuse) vous montrera très vite qu'il n'est pas toujours aisé de retrouver une fausse note dans les datas.

### Coder les notes

Ce choix étant fait, il faut trouver un système de codification des notes. Ici commence la plus grosse difficulté. De ce choix va dépendre toute la lisibilité du programme et surtout des datas. La première chose à faire est de supprimer les valeurs supérieures et infé-

rieures des tables de périodes. Quatre octaves sont largement suffisantes pour interpréter la majeure partie des partitions. De là, chacun doit trouver le système qui lui paraît le plus pratique. La méthode la plus simple, que j'utilise dans cet article, consiste à rentrer directement la valeur de la période. Cette méthode paraît peu pratique au début car il faut regarder à chaque fois la valeur de la note. Avec un peu d'expérience, ce handicap s'estompe et l'on finit par connaître par cœur les valeurs de ces périodes.

Une seconde méthode consiste à coder les notes. L'homme n'étant pas fait pour l'effort, utilisez les douze premières lettres de l'alphabet plutôt que do, do, etc... De



# à Amstrad

*très heureux de pouvoir faire de la musique avec ces jouets magnifiques que sont les synthétiseurs et autres ordinateurs. Alors, à titre posthume, je dédie cet article à celui dont j'ai essayé de traduire une partition en datas.*

plus, il faut toujours penser que la mémoire n'est pas élastique alors autant ménager le nombre de caractères. Si vous désirez travailler de la sorte, n'oubliez pas de coder aussi l'octave dans laquelle se trouve la note.

## Programmer période et durée

Une fois la valeur de la note définie, il reste encore à définir la façon de programmer la durée de celle-ci. Contrairement aux périodes qui sont définies une fois pour toutes, à moins d'une composition personnelle, il me semble intéressant de pouvoir changer le tempo à volonté. Dans ce cas-là, il paraît impossible de rentrer de façon fixe cette durée. De nouveau, deux méthodes de travail vont être possibles. La première consiste à utiliser la première lettre de toutes les valeurs possibles : blanche = B, triple croche = T, etc. Une fois la lecture des datas faite, il faudra alors décoder ces valeurs.

Cette méthode est très bien pour les amateurs de programmation mais me semble un peu un marteau pour écraser une mouche. Quatre-vingt-dix pour cent des partitions prenant la noire pour référence, pourquoi ne pas lui attribuer la valeur de 1. Cette méthode donne donc :



ronde	=4
blanche	=2
noire	=1
croche	=0.5
double croche	=0.25
triple croche	=0.125
note pointée = note	x 1.5

Tout comme la valeur de la période, le choix dépend là de l'utilisateur. Dans le programme illustrant cet article, je me suis contenté de programmer la période et la durée. Il est bien entendu que l'on pourrait faire la même chose avec tous les paramètres de la fonction SOUND. Compte tenu du nombre de paramètres, il est ainsi tout à fait possible de changer le son de chaque note mais pour le morceau choisi ici, cela aurait fait du « psychémozart ». Amusez-vous à écrire

un petit morceau et à faire varier les enveloppes de ton et le numéro du canal. Il y a de quoi se retrouver la tête en bas sans s'en rendre compte.

## Synchroniser

Une autre des difficultés à surmonter va être la synchronisation des voix. Comme je vous l'avais expliqué la fois précédente, il existe sur l'Amstrad de nombreuses possibilités pour synchroniser les trois canaux. Faire démarrer ceux-ci ensemble est très simple mais, cela ne suffit pas toujours. Lorsque l'on veut recopier une partition, deux cas vont se présenter. Dans le premier, les deux ou trois voix partent ensemble et continuent toujours ensemble au même rythme. Alors là, tout est simple. Vous entrez les datas de période et de durée des différentes voix puis vous les jouez avec SOUND 33 sur la première voix, SOUND 34 sur la deuxième voix et enfin SOUND 28 sur la troisième.



Malheureusement, vous allez vite vous lasser de la musique que vous enregistrez : adieu les fugues et autres canons. Si donc vous préférez le contrepoint, (musique où les voix arrivent et repartent dans tous les sens), il va falloir synchroniser les voix sans pour autant qu'elles s'attendent à chaque note. Pour obtenir cela, chaque canal va rester indépendant et les trois voix seront jouées avec SOUND 1, SOUND 2 et SOUND 4. De cette manière, il sera possible d'avoir quatre noires sur la première voix, un silence une noire et quatre croches sur la deuxième voix et enfin une blanche pointée et une noire sur la troisième voix.

Une des applications dont je n'ai pas encore parlé est l'utilisation de l'Amstrad en séquenceur. En effet, rien n'empêche de boucler le programme sur lui-même afin d'obtenir une musique continue. La seule contrainte est l'obligation de lire les datas et de boucler le programme uniquement sur l'interprétation. Sinon attention aux attentes. Par contre, il est tout à fait possible de remonter ou de descendre chaque séquence d'une tierce ou une quinte. Si vous jouez d'un instrument, votre fidèle ordinateur pourra donc vous accompagner. Il en est de même pour une utilisation en boîte à rythme. Le plus difficile ici est de trouver de bons sons de percussions.

## Le programme

J'en viens maintenant au programme que je vous propose. J'ai essayé de concevoir ce



programme comme un programme type. Je pense qu'en ne changeant que les datas et quelques variables, il vous sera possible de jouer n'importe quelle partition à deux voix à partir de ce modèle. Nous allons donc passer en revue les différentes parties de ce listing.

A la ligne 40, j'indique le tempo, nv, vx1 et vx2. Comme j'ai expliqué plus haut, il est préférable de ne pas fixer la durée de chaque note dans les datas. Par contre, fixer en début de programme, de façon claire le tempo, va permettre de le modifier facilement au gré de l'utilisateur. nv correspond au nombre de voix. vx1 et vx2 indiquent le nombre de notes à lire puis à jouer sur chaque voix.

A la ligne 50, j'ouvre mon fichier dimensionné par les variables introduites à la ligne précédente. De cette manière, vous n'aurez

à changer qu'une fois le nombre de voix et de notes lors de vos propres copies de partitions.

Je passe rapidement sur les lignes 70 à 120 où je vous laisse entière liberté de délirer. Pourquoi ne pas ajouter le dessin de la partition ? Mais ce n'est pas le but aujourd'hui. De la ligne 130 à la ligne 140, je vais lire les datas (périodes de durée) de la première voix. En 150 et 160 je répète la même opération mais, cette fois, pour la deuxième voix. Mes datas sont maintenant chargées et je peux en faire ce que je veux. N'étant qu'un simple musicien, eh bien, je vais les transformer en notes de musique.

Tout d'abord, je me souviens que l'Amstrad compte la durée des notes en centièmes de secondes alors que la norme musicale normale est le nombre de noires/minute. Une petite règle de trois à la ligne 170 et le tour

est joué. Ensuite, ma sensibilité musicale me demande de trouver un son agréable pour auditionner mon duo. J'écris donc une enveloppe entre le piano et le clavecin (électronique !). De nouveau, je laisse libre votre imagination pour d'autres enveloppes. Enfin, tout est prêt. Les musiciens, oh pardon le AY8910 a ses datas sur son pupitre. Alors jouez maintenant. Les lignes 200 et 210 vous offrent un concert gratuit (oui je sais, Amstrad Magazine coûte 19 Francs !).

Voilà donc expliqué en quelques lignes comment traduire une partition pour la rendre compréhensible par votre micro-ordinateur. Un seul souhait maintenant, pouvoir un jour organiser un concert où chacun viendrait interpréter son œuvre. Alors avis aux amateurs et bonnes musiques.

*Dominique Gourdier*

```

10 REM DUO DE MOZART
20 REM PAR D. GOURDIER
30 REM VERSION MAI 1986
40 tempo = 120: nv =2: vx1= 72: vx2= 69
50 DIM no(nv,vx1), du(nv,vx1)
60 CLS: MODE 0
70 LOCATE 3,5:PRINT "DUO (MOZART)"
80 LOCATE 3,7:PRINT"INTERPRETE PAR"
90 LOCATE 1,9:PRINT "LE PHILHARMONIQUE"
100 LOCATE 10,11:PRINT "DE"
110 LOCATE 8,15:PRINT"AMSTRAD"
120 LOCATE 3, 20:PRINT"SOLISTE : D.GOURDIER"
130 RESTORE 240
140 FOR i=1 TO vx1: READ no(1,i),du(1,i):NEXT
150 RESTORE 290
160 FOR i=1 TO vx2: READ no(2,i),du(2,i):NEXT
170 tempo = (60*100)/tempo
180 ENV 1,1,10,5,1,-2,5,10,0,5,6,-1,5
190 FOR i=1 TO vx1
200 SOUND 1,no(1,i),du(1,i)*tempo,15,1
210 SOUND 2,no(2,i),du(2,i)*tempo,15,1
220 NEXT
230 END
240 DATA 213,2,253,1,239,1,284,1,284,1,0,1,284,.5,253,.5,239,.5,284,.5,253,1,319
,.5,284,.5,253,.5,239,.5
250 DATA 213,2,253,1,239,1,284,1,284,1,0,1,284,.5,253,.5,239,.5,284,.5,319,1,0,2
,284,2,253,1,284,2,253,1
260 DATA 239,2,253,1,239,2,253,1,225,3,213,1,284,1,284,1,268,2,319,1,338,1,319,.
5,284,.5,253,.5,225,.5
270 DATA 213,2,253,1,239,1,284,1,284,1,0,1,284,.5,253,.5,239,.5,284,.5,253,1,319
,.5,284,.5,253,.5,239,.5
280 DATA 213,2,253,1,239,1,284,1,284,1,0,1,284,.5,253,.5,239,.5,284,.5,319,1
290 DATA 253,2,319,1,284,1,426,1,426,1,426,1,426,1,426,1,319,1,638,1,638,1,638,.
5,638,.5,506,.5,425,.5,319,.5,426,.5
300 DATA 284,1,426,1,426,1,426,1,426,1,426,1,638,1,0,1,0,1,0,3,319,1,
358,1,319,1,319,1,358,1,319,1,284,1,190,1,213
,1
310 DATA 268,1,284,1,319,1,338,1,0,1,426,1,451,3,426,1,0,1,0,1,253,2,319,1,284,1
,426,1,426,1,426,1,426,1,426,1,319,1,638,1,638,1
320 DATA 638,.5,638,.5,506,.5,426,.5,426,.5,284,1,426,1,426,1,426,1,426,1,426,1,
638,1

```